UNITED STATES

DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

FORTRAN COMPUTER PROGRAM FOR CALCULATION OF STRESS-INTENSITY FACTORS, STRESSES, AND DISPLACEMENTS ASSOCIATED WITH A FLUID-PRESSURIZED FRACTURE NEAR THE EARTH'S SURFACE

by

David D. Pollard

Open-File Report No. 78-160

This report is preliminary and has not been edited or reviewed for conformity with Geological Survey standards or nomenclature.

INTRODUCTION

Both natural and man-made extension fractures are found in a wide variety of geological settings near the surface of the earth. These fractures are planar or gently curving discontinuities along which the solid has separated, primarily in directions normal to the discontinuity. Dimensions measured in the fracture plane may range from a few centimeters to more than fifty kilometers. In cross section the maximum width-to-length ratio is small, generally less than 1/100. These fractures are commonly filled with a fluid (e.g. water, oil, or gas) or a solid (e.g. ice, igneous rock, or hydrothermal minerals) believed to have been fluid before it crystallized. The observed association of fluids and extension fractures suggests that fluid pressure and pressure gradients played an important role in dilation and propagation of the fracture. Some natural extension fractures, such as desiccation cracks in mud and cooling joints in lava flows, initiate at the surface and propagate downward. Here we consider fractures that initiate at depth and interact with the surface, but do not intersect it.

This report is a supplement to the paper "On the Mechanical Interaction Between a Fluid-filled Fracture and the Earth's Surface" by David D. Pollard and Gary Holzhausen, Tectonophysics (in press, 1978). Considerable discussion of geological applications, previous work, and details of the method of solution are presented in that paper. In this report we present a FORTRAN computer program for calculating the stress-intensity factors, stresses, and displacements associated with the pressurization of fractures near the earth's surface.

THE COMPUTER PROGRAM

Figure 1 illustrates the model geometry and parameters. In the following discussion, symbols in parentheses are the names of variables and constants used in the FORTRAN program. A slit of half-length a (2*R) and zero initial width is positioned at depth-to-center d (DC) below the horizontal free surface of a semi- infinite, elastic region. The inclination of the slit from the surface is α (DIP). Orthogonal Cartesian coordinates x, y originate at the slit center and are parallel and perpendicular to the slit. A second set of Cartesian coordinates u, v originate on the surface immediately above the slit center and are parallel and perpendicular to the surface. In reference to the slit coordinates stress are σ_{xx}, σ_{xy}, σ_{yy} (XX, XY, YY) and displacements are δ_x, δ_y (DX, DY). In reference to the surface coordinates stresses are σ_{uu}, σ_{uv}, σ_{vv} (UU, UV, VV) and displacements are δ_u, δ_v (DU, DV). Stress and displacement are positive when directed as shown in Figure 1. Stress-intensity factors for the positive end of the slit are K_I^+, K_{II}^+ (SLKP) and for the negative end are K_I^-, K_{II}^- (SLKN). Functions of complex variables are used extensively in the program. We make use of the complex plane z = x + iy (Z) related to the slit and the complex plane w = u + iv (W) related to the surface. The material throughout the semi-infinite region is homogeneous, isotropic, and elastic, and in static equilibrium. The model is two dimensional and conditions of plane strain are specified. The surface of the semi-inifinite region is free of normal and shear stress. The slit wall is free of shear stress but subjected to an arbitrary normal stress (pressure) distribution -p(x). Minor alterations of the program would allow shear stresses to be specified on the slit.


a
Earth's surface
fracture filled with water, air, or magma
b
free surface
v
u
α
σ_{vv}
σ_{uv}
σ_{uu}
δ_v
δ_u
d
x
y
δ_y
δ_x
a
σ_{yy}
σ_{xy}
σ_{xx}
slit

FIGURE 1

There are several different methods, both analytical and numerical, of solving boundary value problems in the theory of elasticity that involve slits. The method used here has been called the Schwarz-Neumann alternating technique, Schwarz algorithm, or simply the method of successive approximations. This method is particularly useful for solving problems in regions bounded by several contours. Analytical solutions for several different regions, each bounded by one of these contours, are successively (iteratively) superimposed. Each new superposition results in the correct boundary conditions on one of the contours but alters the boundary condition on all others. Conditions on these other contours then must be adjusted towards the desired conditions by succeeding superpositions. The process continues until the desired boundary conditions are approximated on all contours. The rapid numerical convergence of the program, the coincidence of the results with analytical solutions for well known special cases, and the coincidence of the results with numerical results obtained by others using different techniques give us considerable confidence in the overall accuracy.

We use two basic analytical solutions because there are two boundaries, the free surface and the slit wall, that define the region of interest. These two solutions are contained in two subroutines called by the main program. The first solution (subroutine SURF) is for the uniform loading of a portion of the planar surface of a semi-infinite region. The second (subroutine SLIT) is for the uniform loading of a portion of the wall of a slit in an infinite region. Both regions have the point at infinity in common and the same boundary condition of zero stress at this point. The two solutions are used to approximate non-uniform distributions of normal and tangential loading on the surface and on the slit. By making the intervals of uniform loading sufficiently small and by superimposing solutions for different loadings on adjacent intervals, a good approximation of any non-uniform distribution is achieved.

The procedure for successively superimposing the analytical solutions is illustrated in Figure 2. Arrows in this figure are only a schematic representation of stresses at isolated points on the slit and surface. They do not necessarily represent the magnitudes, signs, or distributions of stresses in actual solutions. The basic steps in the procedure are:

a) The desired normal stress distribution n(x) is approximated on the slit boundary in the finite region by superimposing different loads on numerous adjacent intervals of the slit wall. The resulting stress state and displacements referred to x, y coordinates at numerous stations along the slit and the line v = 0 are calculated.

b) The stress and displacement components on the line v = 0 are transformed to refer to u, v coordinates. This gives the normal stress $\sigma_{vv} = n_1(u)$ and tangential stress $\sigma_{uv} = t_1(u)$ stress distributions on the line v = 0 in the infinite region.

c) The normal stress distribution $-n_1(u)$ and tangential stress distribution $-t_1(u)$ distributions are approximated on the surface of the semi-infinite region by applying different uniform loads on numerous adjacent intervals. The resulting stress state and displacements referred to u, v coordinates are calculated at numerous stations on the surface and the line $|x| \leq a$, y = 0.

d) The stress and displacement components on the line $|x| \leq a$, y = 0 are transformed to refer to the x, y coordinates. This gives the normal stress $\sigma_{xy} = n_1(x)$ and tangential stress $\sigma_{xy} = t_1(x)$ distributions on this line in the semi-infinite region.


a
σvv = n1(u)
σuv = t1(u)
x
y
n(x)
v
-n1(u)
-t1(u)
u
σxy=Δt1(x)
σyy=Δn1(x)
free surface
Δt1(x)
n(x)+Δn1(x)
b
free surface
Δt1(x)
n(x)+Δn1(x)
σvv = n2(u)
σuv = t2(u)
x
y
-Δt1(x)
-Δn1(x)
v
-n2(u)
-t2(u)
u
σxy=Δt2(x)
σyy=Δn2(x)
free surface
Δt2(x)
n(x)+Δn2(x)

FIGURE 2

e) Summation of all solutions considered thus far results in a free surface along $v = 0$ as desired. However, on the slit the desired stress distribution is altered by the excess stress $n_1(x)$, $t_1(x)$.

f) If the excess stresses are sufficiently small relative to the desired stress $n(x)$, the problem is solved. If not, steps a) to e) are successively repeated using the negative of the excess stress $-n_1(x)$, $-t_1(x)$ for the load on the slit in step a).

Four arbitrary parameters (ERSL, ERSU, ERDP, TDS) control the precision of the final results. The interval length on the slit is adjusted using ERSL, and the interval length on the surface is adjusted using ERSU. As ERSL and ERSU is decreased, so is the interval length. The lengths are not uniform, but vary inversely with the stress gradient on the slit or surface. The acceptable magnitude of excess stress is adjusted using ERDP. As ERDP is decreased, the number of iterations (NIT) is increased and the precision of the boundary conditions is increased. The total distance considered on the surface is defined as TDS.

The output is normalized by factors supplied as data. The stresses are normalized by SNORM, displacements by DNORM and stress intensities by KNORM. Although the method of solution allows one to calculate stresses and displacements anywhere in the region, we only calculate the stress-intensity factors at the ends of the slit, the stresses and displacements of the slit wall, and the stresses and displacements of the free surface. Variations in these stresses and displacements are related most easily to practical problems.

The computer program is listed on the following pages. We have included numerous comment statements within the program to supplement the preceeding remarks. The output from two runs of the program are included to aid users in checking their results. The program may be adapted easily to solve the problem of a slit with uniform or linearly varying shear stress. To do this, replace the lines marked with asteriks in the right margin with the lines listed at the end of the program. The author of this report would be pleased to offer assistance with further interpretations of the program, and its results. Please call (415) 323-8111, ext. 2635.

REFERENCES

Carrier, G. F., Krook, M., and Pearson, C. E., 1966, Functions of a Complex Variable, McGraw-Hill Book Co., New York, 438 p.

Muskhelishvili, N. I., 1975, Some Basic Problems of the Mathematical Theory of Elasticity, Noordhoff International Publishing, Leyden, Netherlands, 703 p.

Pollard, D. D., and Holzhausen, G., 1978, On the mechanical interaction between a fluid-filled fracture and the earth's surface, Tectonophysics (in press).

Sokolnikoff, I. S., 1956, Mathematical Theory of Elasticity, McGraw-Hill Book Co., New York, 476 p.

```
C
C     CALCULATE THE TWO-DIMENSIONAL, LINEAR ELASTIC (PLAIN STRAIN)
C     DISPLACEMENTS, STRESSES, AND STRESS INTENSITIES ASSOCIATED
C     WITH A SLIT SUBJECTED TO AN ARBITRARY PRESSURE DISTRIBUTION
C     AND ORIENTED AT AN ARBITRARY ANGLE TO A NEARBY FREE SURFACE.
C     USE THE SCHWARZ-NEUMANN ALTERNATING METHOD OF SOLUTION
C     (SOKOLNIKOFF, 1956, PP. 318-326; MUSKHELISHVILI, 1975,
C     PP. 389-390). HERE WE IMPOSE A UNIFORM OR LINEARLY VARYING
C     PRESSURE DISTRIBUTION.
C
      DIMENSION PD(5000),SD(5000),RWS(5000),DPD(500),RZS(500),
     2          DPI(100),DSI(100),DP(100),DS(100)
      REAL*4 KNORM,XX(100)/100*0./,YY(100)/100*0./,XY(100)/100*0./,
     2       DX(100)/100*0./,DY(100)/100*0./,
     3       XXN(100)/100*0./,DXN(100)/100*0./,DYN(100)/100*0./,
     4       UU(500)/500*0./,VV(500)/500*0./,UV(500)/500*0./,
     5       DU(500)/500*0./,DV(500)/500*0./,TILT(500)/500*0./
      COMPLEX W0,CDIP,Z1,Z2,Z,SXY2,DXY,KP,KN,WS(1002),ZS(202),
     2        WI1,WI2,W,W1,W2,SUV2,DUV,SLKP,SLKN,ZF,
     3        CMPLX,CEXP,CONJG
    2 FORMAT(5G12.4)
   10 FORMAT(1H0,' R = ',G12.4,' DC = ',G12.4,' DIP = ',G12.4)
   12 FORMAT(1H0,' SM = ',G12.4,' PR = ',G12.4,' DPC = ',G12.4,
     2       ' DPG = ',G12.4)
   14 FORMAT(1H0,' TDS = ',G12.4,' ERSU = ',G12.4,' ERSL = ',G12.4,
     2       ' ERDP = ',G12.4)
   15 FORMAT(1H0,' KNORM = ',G12.4,' SNORM = ',G12.4,' DNORM = ',
     2       G12.4)
   16 FORMAT(1H0,' NSURF = ',I6)
   18 FORMAT(1H0,' NSLIT = ',I6)
   20 FORMAT(1H0,' NIT = ',I6,' NSUPER = ',I10)
   22 FORMAT(1H0,' KP = ',2G12.4,' KN = ',2G12.4)
   24 FORMAT(1H0,'      X                XX/XXN          DX/DXN          DY/DYN
     2          YY/XY')
   26 FORMAT(1H0,5G15.4)
   27 FORMAT(1H0,15X,4G15.4)
   28 FORMAT(1H0,'       U                UU              DU              DV
     2')
C
C      SUPPLY THE FOLLOWING DATA:
C            R = .25*SLIT LENGTH
C            DC = PERPENDICULAR DISTANCE FROM SURFACE TO SLIT CENTER
C            DIP = INCLINATION OF SLIT FROM SURFACE (DEGREES)
C            SM = SHEAR MODULUS
C            PR = POISSON'S RATIO
C            DPC = DRIVING PRESSURE AT SLIT CENTER                   *1
C            DPG = DRIVING PRESSURE GRADIENT                         *2
C            TDS = TOTAL DISTANCE ON SURFACE (EG. 20*DC)
C            ERSU = ADJUSTMENT FACTOR FOR STATIONS ON SURFACE (EG. .01)
C            ERSL = ADJUSTMENT FACTOR FOR STATIONS ON SLIT (EG. .01)
C            ERDP = ADJUSTMENT FACTOR FOR ITERATIONS (EG. .01)
C            KNORM = NORMALIZATION FACTOR FOR STRESS INTENSITY
C            SNORM = NORMALIZATION FACTOR FOR STRESS
C            DNORM = NORMALIZATION FACTOR FOR DISPLACEMENT
C
       READ(5,2)R,DC,DIP,SM,PR
       READ(5,2)DPC,DPG,TDS
       READ(5,2)ERSU,ERSL,ERDP
       READ(5,2)KNORM,SNORM,DNORM
```

```
      WRITE(6,10)R,DC,DIP
      WRITE(6,12)SM,PR,DPC,DPG
      WRITE(6,14)TDS,ERSU,ERSL,ERDP
      WRITE(6,15)KNORM,SNORM,DNORM
C
C     CALCULATE COORDINATES WS(I) OF STATIONS ON SURFACE.
C     STATION SPACING VARIES INVERSELY WITH THE VARIATION OF
C     MEAN NORMAL STRESS P AND MAXIMUM SHEAR STRESS S ON THE
C     SURFACE DUE TO UNIFORM PRESSURIZATION OF THE SLIT.
C     CALCUALTE ABSOLUTE DIFFERENCE FOR P AND S BETWEEN 5000
C     UNIFORMLY DISTRIBUTED ADJACENT POINTS OVER DISTANCE TDS.
C     ADD DIFFERENCES FOR P AND S RESPECTIVELY FOR ADJACENT
C     INTERVALS UNTIL EITHER TOTAL JUST EXCEEDS THE PRODUCT ERSU*DPC.
C     USE THE SUMMED INTERVAL DISTANCE AS STATION SPACING.
C     ADJUST SPACING FOR SYMMETRIC GEOMETRY TO BE SYMMETRIC.
C
      ZERO=0.
      UWO=-DC
      WO=CMPLX(ZERO,UWO)
      UDIP=(DIP/360.)*6.283185
      CDIP=CMPLX(ZERO,UDIP)
      RZ1=2.*R
      Z1=CMPLX(RZ1,ZERO)
      Z2=-Z1
      WS1=.5*TDS
      RWS(1)=WS1
      Z=(WS1-WO)*CEXP(-CDIP)
      CALL SLIT(Z,Z1,Z2,R,DPC,ZERO,SM,PR,P1,SXY2,DXY,KP,KN)
      S1=CABS(SXY2)
      DO 110 I=2,5000
      WS2=WS1-TDS/5000.
      Z=(WS2-WO)*CEXP(-CDIP)
      CALL SLIT(Z,Z1,Z2,R,DPC,ZERO,SM,PR,P2,SXY2,DXY,KP,KN)
      S2=CABS(SXY2)
      PD(I)=ABS(P1-P2)
      SD(I)=ABS(S1-S2)
      RWS(I)=WS2
      WS1=WS2
      P1=P2
      S1=S2
  110 CONTINUE
      WS(1)=CMPLX(RWS(1),ZERO)
      NSURF=2
      PDT=0.
      SDT=0.
      DPER=ABS(DPC)*ERSU
      DO 114 I=2,5000
      PDT=PDT+PD(I)
      SDT=SDT+SD(I)
      IF(PDT.LT.DPER.AND.SDT.LT.DPER) GO TO 114
      WS(NSURF)=CMPLX(RWS(I),ZERO)
      NSURF=NSURF+1
      PDT=0.
      SDT=0.
  114 CONTINUE
      WS(NSURF)=-WS(1)
      IF(DIP.NE.90..AND.DIP.NE.0.) GO TO 119
      DO 116 I=1,NSURF
      NSUH=I-1
      IF(REAL(WS(I)).LT.0.) GO TO 117
```

```
  116 CONTINUE
  117 NSURF=2*NSUH
      DO 118 I=1,NSUH
  118 WS(NSUH+I)=-WS(NSUH+1-I)
  119 WRITE(6,16)NSURF
C
C     CALCULATE COORDINATES ZS(I) OF STATIONS ON SLIT.
C     STATION SPACING VARIES INVERSELY WITH THE VARIATION OF
C     MEAN NORMAL STRESS P AND MAXIMUM SHEAR STRESS S ON THE
C     IMAGE SLIT DUE TO UNIFORM PRESSURIZATION OF THE SLIT.
C     IT ALSO VARIES INVERSLY WITH THE VARIATION OF PRESSURE DP.
C     CALCULATE ABSOLUTE DIFFERENCE FOR P AND S BETWEEN 500
C     UNIFORMLY DISTRIBUTED ADJACENT POINTS ON IMAGE SLIT AND
C     ABSOLUTE DIFFERENCE FOR DP BETWEEN 500 POINTS ON SLIT.
C     ADD DIFFERENCES FOR P, S, AND DP RESPECTIVELY FOR ADJACENT
C     INTERVALS UNTIL ANY TOTAL JUST EXCEEDS THE PRODUCT ERSL*DPC.
C     USE SUMMED INTERVAL DISTANCE AS STATION SPACING.
C     ADJUST SPACING FOR SYMMETRIC GEOMETRIES TO BE SYMMETRIC.
C
      ZS1=2.*R
      RZS(1)=ZS1
      WI1=CONJG(ZS1*CEXP(CDIP)+W0)
      Z=(WI1-W0)*CEXP(-CDIP)
      CALL SLIT(Z,Z1,Z2,R,DPC,ZERO,SM,PR,P1,SXY2,DXY,KP,KN)
      S1=CABS(SXY2)
      DP1=DPC+DPG*ZS1
      DO 120 I=2,50
      ZS2=ZS1-4.*R/50.
      WI2=CONJG(ZS2*CEXP(CDIP)+W0)
      Z=(WI2-W0)*CEXP(-CDIP)
      CALL SLIT(Z,Z1,Z2,R,DPC,ZERO,SM,PR,P2,SXY2,DXY,KP,KN)
      S2=CABS(SXY2)
      DP2=DPC+DPG*ZS2
      PD(I)=ABS(P1-P2)
      SD(I)=ABS(S1-S2)
      DPD(I)=ABS(DP1-DP2)
      RZS(I)=ZS2
      ZS1=ZS2
      P1=P2
      S1=S2
      DP1=DP2
  120 CONTINUE
      ZS(1)=CMPLX(RZS(1),ZERO)
      NSLIT=2
      PDT=0.
      SDT=0.
      DPDT=0.
      DPER=ABS(DPC)*ERSL
      DO 124 I=2,50
      PDT=PDT+PD(I)
      SDT=SDT+SD(I)
      DPDT=DPDT+DPD(I)
      IF(PDT.LT.DPER.AND.SDT.LT.DPER.AND.DPDT.LT.DPER) GO TO 124
      ZS(NSLIT)=CMPLX(RZS(I),ZERO)
      NSLIT=NSLIT+1
      PDT=0.
      SDT=0.
      DPDT=0.
  124 CONTINUE
      ZS(NSLIT)=-ZS(1)
```

```
      IF(DIP.NE.0.) GO TO 129
      DO 126 I=1,NSLIT
      NSLH=I-1
      IF(REAL(ZS(I)).LT.0.) GO TO 127
  126 CONTINUE
  127 NSLIT=2*NSLH
      DO 128 I=1,NSLH
  128 ZS(NSLH+I)=-ZS(NSLH+1-I)
  129 WRITE(6,18)NSLIT
C
C     DEFINE THE DESIRED BOUNDARY STRESSES ON THE SLIT.
C     THESE MAY BE ALTERED TO SUIT THE PROBLEM AT HAND.
C     CALCULATE NORMAL BOUNDARY STRESS DPI ON SLIT DUE TO AVERAGE
C     DRIVING PRESSURE AND GRADIENT. SET TANGENTIAL BOUNDARY STRESS
C     TO ZERO.
C
      NSLM1=NSLIT-1
      NSUM1=NSURF-1
      DO 130 I=1,NSLM1                                                   *3
      DPI(I)=DPC+DPG*REAL(.5*(ZS(I)+ZS(I+1)))                            *4
      DP(I)=DPI(I)
      DSI(I)=0.
      DS(I)=DSI(I)
  130 CONTINUE
C
C     CALCULATE STRESSES AND DISPLACEMENTS ON SLIT AND SURFACE
C     DUE TO NORMAL AND TANGENTIAL STRESSES DP;DS ON THE SLIT.
C     ON SLIT STRESSES ARE XX,YY,XY, AND DISPLACEMENTS ARE DX,DY.
C     ON THE OPPOSITE SLIT WALL CALCULATE XXN,DXN,AND DYN.
C     ON SURFACE STRESSES ARE UU,VV,UV, AND DISPLACEMENTS ARE DU,DV.
C     CALCULATE STRESS INTENSITY FACTORS AT POSITIVE SLKP AND
C     NEGATIVE SLKN TIPS OF SLIT DUE TO BOUNDARY LOADING.
C     RETURN TO 132 TO REMOVE RESIDUAL STRESSES FROM SLIT.
C
      NIT=0
      NSUPER=0
      UZF=R*1.E-10
      ZF=CMPLX(ZERO,UZF)
      SLKP=CMPLX(ZERO,ZERO)
      SLKN=CMPLX(ZERO,ZERO)
  132 NIT=NIT+1
      DO 144 I=1,NSLM1
      Z1=ZS(I)
      Z2=ZS(I+1)
      DO 140 J=1,NSLM1
      Z=.5*(ZS(J)+ZS(J+1))
      CALL SLIT(Z,Z1,Z2,R,DP(I),DS(I),SM,PR,SXY1,SXY2,DXY,KP,KN)
      NSUPER=NSUPER+1
      XX(J)=XX(J)+.5*(SXY1-REAL(SXY2))
      YY(J)=YY(J)+.5*(SXY1+REAL(SXY2))
      XY(J)=XY(J)+.5*AIMAG(SXY2)
      DX(J)=DX(J)+REAL(DXY)
      DY(J)=DY(J)+AIMAG(DXY)
  140 CONTINUE
      DO 141 J=1,NSLM1
      Z=.5*(ZS(J)+ZS(J+1))-ZF
      CALL SLIT(Z,Z1,Z2,R,DP(I),DS(I),SM,PR,SXY1,SXY2,DXY,KP,KN)
      NSUPER=NSUPER+1
      XXN(J)=XXN(J)+.5*(SXY1-REAL(SXY2))
      DXN(J)=DXN(J)+REAL(DXY)
```

```
      DYN(J)=DYN(J)+AIMAG(DXY)
  141 CONTINUE
      DO 142 J=1,NSUM1
      W=.5*(WS(J)+WS(J+1))
      Z=(W-W0)*CEXP(-CDIP)
      CALL SLIT(Z,Z1,Z2,R,DP(I),DS(I),SM,PR,SXY1,SXY2,DXY,KP,KN)
      NSUPER=NSUPER+1
      SUV1=SXY1
      SUV2=SXY2*CEXP(-2.*CDIP)
      DUV=DXY*CEXP(CDIP)
      UU(J)=UU(J)+.5*(SUV1-REAL(SUV2))
      VV(J)=VV(J)+.5*(SUV1+REAL(SUV2))
      UV(J)=UV(J)+.5*AIMAG(SUV2)
      DU(J)=DU(J)+REAL(DUV)
      DV(J)=DV(J)+AIMAG(DUV)
  142 CONTINUE
      SLKP=SLKP+KP
      SLKN=SLKN+KN
  144 CONTINUE
C
C     REMOVE STRESSES VV AND UV FROM SURFACE AND CALCULATE
C     RESULTANT STRESSES AND DISPLACEMENTS ON SURFACE AND SLIT.
C
      DO 154 I=1,NSUM1
      W1=WS(I)
      W2=WS(I+1)
      SUN=-VV(I)
      SUT=-UV(I)
      DO 150 J=1,NSUM1
      W=.5*(WS(J)+WS(J+1))
      CALL SURF(W,W1,W2,SUN,SUT,SM,PR,W0,SUV1,SUV2,DUV)
      NSUPER=NSUPER+1
      UU(J)=UU(J)+.5*(SUV1-REAL(SUV2))
      VV(J)=VV(J)+.5*(SUV1+REAL(SUV2))
      UV(J)=UV(J)+.5*AIMAG(SUV2)
      DU(J)=DU(J)+REAL(DUV)
      DV(J)=DV(J)+AIMAG(DUV)
  150 CONTINUE
      DO 152 J=1,NSLM1
      Z=.5*(ZS(J)+ZS(J+1))
      W=Z*CEXP(CDIP)+W0
      CALL SURF(W,W1,W2,SUN,SUT,SM,PR,W0,SUV1,SUV2,DUV)
      NSUPER=NSUPER+1
      SXY1=SUV1
      SXY2=SUV2*CEXP(2.*CDIP)
      DXY=DUV*CEXP(-CDIP)
      XX(J)=XX(J)+.5*(SXY1-REAL(SXY2))
      XXN(J)=XXN(J)+.5*(SXY1-REAL(SXY2))
      YY(J)=YY(J)+.5*(SXY1+REAL(SXY2))
      XY(J)=XY(J)+.5*AIMAG(SXY2)
      DX(J)=DX(J)+REAL(DXY)
      DXN(J)=DXN(J)+REAL(DXY)
      DY(J)=DY(J)+AIMAG(DXY)
      DYN(J)=DYN(J)+AIMAG(DXY)
  152 CONTINUE
  154 CONTINUE
C
C     COMPUTE NORMAL AND TANGENTIAL STRESSES DP;DS NECESSARY
C     TO RETURN TO DESIRED  BOUNDARY CONDITIONS ON SLIT.
C     IF, AT ANY STATION, THESE STRESSES EXCEED THE PRODUCT OF THE
```

```
C       DESIRED BOUNDARY STRESS TIMES ERDP, ITERATE BACK THROUGH
C       132 TO DECREASE THE MAGNITUDES OF DP AND DS.
C
        DO 160 I=1,NSLM1
        DP(I)=DPI(I)-YY(I)
        DS(I)=DSI(I)-XY(I)
  160   CONTINUE
        DO 162 I=1,NSLM1
        IF((ABS(DP(I)).GT.ABS(DPI(I)*ERDP)).OR.                              *5
     2     (ABS(DS(I)).GT.ABS(DPI(I)*ERDP))) GO TO 132                       *6
  162   CONTINUE
C
C       OUTPUT IS NORMALIZED BY CHARACTERISTIC VALUES AND LISTED.
C       NIT IS THE NUMBER OF INTERATIONS.
C       NSUPER IS THE NUMBER OF SUPERPOSITIONS OF SOLUTIONS.
C
        WRITE(6,20)NIT,NSUPER
        SLKP=SLKP/KNORM
        SLKN=SLKN/KNORM
        WRITE(6,22)SLKP,SLKN
        WRITE(6,24)
        DO 172 I=1,NSLM1
        X=REAL(.5*(ZS(I)+ZS(I+1)))
        XX(I)=XX(I)/SNORM
        XXN(I)=XXN(I)/SNORM
        DX(I)=DX(I)/DNORM
        DXN(I)=DXN(I)/DNORM
        DY(I)=DY(I)/DNORM
        DYN(I)=DYN(I)/DNORM
        WRITE(6,26)X,XX(I),DX(I),DY(I),YY(I)
        WRITE(6,27)XXN(I),DXN(I),DYN(I),XY(I)
  172   CONTINUE
        WRITE(6,28)
        DO 174 I=1,NSUM1
        U=REAL(.5*(WS(I)+WS(I+1)))
        UU(I)=UU(I)/SNORM
        DU(I)=DU(I)/DNORM
        DV(I)=DV(I)/DNORM
        WRITE(6,26)U,UU(I),DU(I),DV(I)
  174   CONTINUE
        STOP
        END
C
C       CALCULATE THE X AND Y COMPONENTS OF STRESS AND DISPLACEMENT
C       AT A POINT Z=X+I*Y NEAR A SLIT IN AN INFINITE REGION
C       SUBJECT TO UNIFORM NORMAL STRESS (SLN) AND UNIFORM TANGENTIAL
C       STRESS (SLT) BOTH ACTING ON OPPOSING PORTIONS OF THE SLIT WALL.
C       STRESS FUNCTIONS ARE DERIVED FOLLOWING MUSKHELISHVILI (1975,
C       PP. 347-358). STRESSES AND DISPLACEMENTS ARE CALCULATED USING
C       THE KOLOSOFF FORMULAE (MUSKHELISHVILI, 1975, PP. 157-158).
C       CALCULATE SLIT TIP STRESS INTENSITY FACTOR.
C       STRESS INTENSITY FACTORS ARE CALCULATED USING THE FIRST STRESS
C       FUNCITION AND THE RELATION OF PARIS AND SIH (1965, PP. 37).
C       THE END POINTS OF THE LOADED PORTION ARE Z1;Z2. THE SLIT
C       IS CENTERED AT THE ORIGIN; LIES PARALLEL TO THE X-AXIS WHICH
C       IS POSITIVE TO THE RIGHT WITH THE Y-AXIS POSITIVE UPWARD;
C       AND THE SLIT LENGTH IS 4*R.
C       PR IS POISSON'S RATIO; SM IS THE SHEAR MODULUS.
C       STRESSES ARE (XX,YY,XY); DISPLACEMENTS ARE (DX,DY).
C       SIGNS OF STRESSES AND DISPLACEMENTS FOLLOW THE TENSOR RULE.
```

```
C     KP;KN ARE STRESS INTENSITY FACTORS AT Z=+2R;-2R.
C
      SUBROUTINE SLIT(Z,Z1,Z2,R,SLN,SLT,SM,PR,SXY1,SXY2,DXY,KP,KN)
      COMPLEX Z,Z1,Z2,ZT,ZTP2,ZTM1,ZTM2,ZTM3,ZTT1,ZTT2,ZTT3,ZTT4,
     2        ZT1,ZT2,ZT1T1,ZT2T1,ZT1T2,ZT2T2,ZT1T3,ZT2T3,
     3        ZT1T4,ZT2T4,ZT12T1,ZT12T2,CNT,CN,OM,OMC,OMP,OMPP,
     4        PH,PHP,PHPP,PS,PSP,SXY2,DXY,KP,KN,ZETA,
     5        CLOG,CONJG,CMPLX
      ZT=ZETA(Z,R)
      ZTP2=ZT*ZT
      ZTM1=1./ZT
      ZTM2=ZTM1*ZTM1
      ZTM3=ZTM2*ZTM1
      ZTT1=ZT+ZTM1
      ZTT2=ZTP2+1.
      ZTT3=ZTP2-1.
      ZTT4=1.-ZTM2
      ZT1=ZETA(Z1,R)
      ZT2=ZETA(Z2,R)
      ZT1T1=ZT1+1./ZT1
      ZT2T1=ZT2+1./ZT2
      ZT1T2=ZT1-1./ZT1
      ZT2T2=ZT2-1./ZT2
      ZT1T3=(ZT1-ZT)*(1./ZT1-ZT)
      ZT2T3=(ZT2-ZT)*(1./ZT2-ZT)
      ZT1T4=CLOG((ZT1-ZT)/(1./ZT1-ZT))
      ZT2T4=CLOG((ZT2-ZT)/(1./ZT2-ZT))
      ZT12T1=ZT1-(1./ZT1)-ZT2+(1./ZT2)
      ZT12T2=CLOG(ZT2/ZT1)
      CHI=3.-4.*PR
      CNT=-R*CMPLX(SLT,SLN)/6.283185
      CT=REAL(CNT)
      RCN=0.
      UCN=AIMAG(CNT)
      CN=CMPLX(RCN,UCN)
      OM=Z
      OMC=CONJG(OM)
      OMP=R*(1.-ZTM2)
      OMPP=2.*R*ZTM3
      PH=CNT*(2.*ZTM1*ZT12T2+(ZTT1-ZT1T1)*ZT1T4-(ZTT1-ZT2T1)*ZT2T4)
      PHP=CNT*(-2.*ZTM2*ZT12T2+ZTM1*ZT12T1+ZTT4*(ZT1T4-ZT2T4))
      PHPP=CNT*(2.*ZTM3*(2.*ZT12T2+ZT1T4-ZT2T4)-ZTM2*ZT12T1
     2          +ZTT4*((ZT1T2/ZT1T3)-(ZT2T2/ZT2T3)))
      PS=CT*(4.*ZTM1*ZT12T2/ZTT3-ZT12T1*ZTT2/ZTT3
     2        -(2.*ZTT1-ZT1T1)*ZT1T4+(2.*ZTT1-ZT2T1)*ZT2T4)
     3   +CN*(4.*ZT/ZTT3*ZT12T2-ZT12T1*ZTT2/ZTT3-ZT1T1*ZT1T4
     4        +ZT2T1*ZT2T4)
      PSP=CT*(4.*(ZTM2-3.)*ZT12T2/ZTT3**2+4.*ZT*ZT12T1/ZTT3**2
     2         +2.*ZTT4*(ZT2T4-ZT1T4)-(2.*ZTT1-ZT1T1)*ZT1T2/ZT1T3
     3         +(2.*ZTT1-ZT2T1)*ZT2T2/ZT2T3)
     4   +CN*(-4.*ZTT2*ZT12T2/ZTT3**2+4.*ZT*ZT12T1/ZTT3**2
     5            -ZT1T1*ZT1T2/ZT1T3+ZT2T1*ZT2T2/ZT2T3)
      SXY1=4.*REAL(PHP/OMP)
      SXY2=2.*(OMC*(OMP*PHPP-OMPP*PHP)/OMP**3+PSP/OMP)
      DXY=.5*(CHI*PH-OM*CONJG(PHP/OMP)-CONJG(PS))/SM
      KP=2.*SQRT(.5/R)*CONJG(CNT*(-2.*ZT12T2+ZT12T1))
      KN=2.*SQRT(.5/R)*CONJG(CNT*(-2.*ZT12T2-ZT12T1))
      RETURN
      END
C
```

```
C      GIVEN THE COMPLEX CARTESIAN COORDINATE Z = X + I*Y OF
C      A POINT, TRANSFORM TO THE COMPLEX POLAR COORDINATE
C      ZETA = RHO*EXP(I*THETA). USE THE JOUKOWSKY TRANSFORMATION
C      (CARRIER, KROOK, AND PEARSON, 1966, P. 157) WHICH
C      TRANSFORMS THE REGION OUTSIDE A SLIT IN THE Z PLANE
C      TO THE REGION OUTSIDE THE UNIT CIRCLE IN THE ZETA PLANE.
C      THE SLIT IS CENTERED AT THE ORIGIN IN THE Z PLANE; LIES
C      ALONG THE X AXIS; AND HAS A TOTAL LENGTH OF 4*R.
C      THE UNIT CIRCLE IN CENTERED AT THE ORIGIN IN THE ZETA PLANE
C      AND THETA IS MEASURED COUNTERCLOCKWISE FROM THE X AXIS.
C      CHOICE OF SIGN IN THE TRANSFORMATION EQUATION DEPENDS UPON
C      THE POINT IN QUESTION AND THE INTERPRETATION OF THE DOUBLE
C      VALUED SQUARE ROOT FUNCTION. THE FOLLOWING CHOICES CONFORM TO
C      THE STANFORD FORTRAN WATFIV COMPILER. THESE SHOULD BE CHECKED
C      ON ANY OTHER SYSTEM.
C
       FUNCTION ZETA(Z,R)
       COMPLEX ZETA,Z,ZDR,CSQRT,CMPLX
       RZ=REAL(Z)
       UZ=AIMAG(Z)
       ZDR=Z/R
       IF(RZ.GE.0.)ZETA=.5*(ZDR+CSQRT(ZDR**2-4.))
       IF(RZ.LT.0.)ZETA=.5*(ZDR-CSQRT(ZDR**2-4.))
       IF(RZ.GT.(-2.*R).AND.UZ.EQ.0.)ZETA=.5*(ZDR+CSQRT(ZDR**2-4.))
       IF(RZ.EQ.0..AND.UZ.LT.0.)ZETA=.5*(ZDR-CSQRT(ZDR**2-4.))
       RETURN
       END
C
C      CALCULATE THE U AND V COMPONENTS OF STRESS AND DISPLACEMENT
C      AT A POINT W=U+I*V IN A SEMI-INFINITE REGION SUBJECTED TO
C      UNIFORM NORMAL STRESS SUN AND A UNIFORM TANGENTIAL STRESS SUT
C      BOTH ACTING OVER A PORTION OF THE OTHERWISE FREE SURFACE.
C      STRESS FUNCTIONS ARE DERIVED FOLLOWING MUSKHELISHVILI (1975,
C      PP. 402-408).
C      THE END POINTS OF THE APPLIED STRESS ARE W1 AND W2.
C      PR IS POISSON'S RATIO; SM IS THE SHEAR MODULUS.
C      STRESSES ARE (UU,VV,UV); DISPLACEMENTS ARE (DU,DV).
C      SIGNS FOR STRESSES AND DISPLACEMENTS FOLLOW TENSOR RULE.
C      ORIGIN IS ON SURFACE OF REGION WITH U-AXIS POSITIVE TO RIGHT
C      AND V-AXIS POSITIVE UPWARD AWAY FROM REGION.
       SUBROUTINE SURF(W,W1,W2,SUN,SUT,SM,PR,W0,SUV1,SUV2,DUV)
       COMPLEX W,W1,W2,WF,WC,S,SC,D,CL1,CL2,WT1,WT2,WT3,WT4,
     2        CPH,CPHP,CPS,PH,PHP,PS,SUV2,DUV,W0,CONJG,CMPLX,CLOG
       RWF=0.
       UWF=1.E-10
       WF=CMPLX(RWF,UWF)
       IF((AIMAG(W).EQ.0.)) W=W-WF
       CHI=3.-4.*PR
       WC=CONJG(W)
       S=CMPLX(SUN,SUT)
       SC=CONJG(S)
       RD=0.
       UD=6.283185
       D=CMPLX(RD,UD)
       CL1=CLOG(W-W1)
       CL2=CLOG(W-W2)
       WT1=CL1-CL2
       WT2=1./(W-W1)-1./(W-W2)
       WT3=(W-W1)*CL1-(W-W2)*CL2
       WT4=W1*CL1-W2*CL2
```

```
      CPH=-SC*WT1/D
      CPHP=-SC*WT2/D
      CPS=((SC-S)*WT1+SC*W*WT2)/D
      PH=-SC*WT3/D
      PHP=CPH
      PS=((SC-S)*WT3+SC*WT4)/D
      SUV1=4.*REAL(CPH)
      SUV2=2.*(WC*CPHP+CPS)
      DUV=.5*(CHI*PH-W*CONJG(PHP)-CONJG(PS))/SM
      CL1=CLOG(W0-W1)
      CL2=CLOG(W0-W2)
      WT1=CL1-CL2
      WT3=(W0-W1)*CL1-(W0-W2)*CL2
      WT4=W1*CL1-W2*CL2
      PH=-SC*WT3/D
      PHP=-SC*WT1/D
      PS=((SC-S)*WT3+SC*WT4)/D
      DUV=DUV-.5*(CHI*PH-W0*CONJG(PHP)-CONJG(PS))/SM
      RETURN
      END
C
C       INPUT DATA IS LISTED BELOW FOR THE FIRST EXAMPLE OUTPUT.
C
   .5000       1.5000      90.0000    .1000E+06        .2500
 -1.0000       0.0000      30.0000
  0.0200       0.0100       0.0100
  1.0000       1.0000     .7500E-06
C
C       REPLACE THE INDICATED LINES WITH THE FOLLOWING IN ORDER
C       TO LOAD THE CRACK IN SHEAR.
C
C               DPC = DRIVING SHEAR STRESS AT SLIT CENTER                  *1
C               DPG = DRIVING GRADIENT OF SHEAR                            *2
       DPI(I)=0.                                                           *3
       DSI(I)=DPC+DPG*REAL(.5*(ZS(I)+ZS(I+1)))                             *4
       IF((ABS(DP(I)).GT.ABS(DSI(I)*ERDP)).OR.                             *5
      2   (ABS(DS(I)).GT.(DSI(I)*ERDP))) GO TO 132                         *6
```

```
R =      0.5000      DC =      1.500      DIP =      90.00      DPG =      0.0000
SM =     0.1000E 06  PR =      0.2500     DPC =      -1.000     ERDP =     0.1000E-01
TDS =    30.00       ERSU =    0.2000E-01 ERSL =     0.1000E-01
KNORM =  1.000       SNORM =   1.000      DNORM =    0.7500E-06
NSURF =  110
NSLIT =  21
NIT =    3  NSUPER =  51123                          -0.9434E-06
KP =     1.198       -0.8080E-05  KN =    1.094
```

X	XX/XXN	DX/DXN	DY/DYN	YY/XY
0.9800	-1.427	-4.462	2.381	-0.9958
	-1.427	-4.462	-2.381	0.3307E-07
0.9400	-1.406	-4.246	4.074	-0.9959
	-1.406	-4.246	-4.074	0.8265E-07
0.9000	-1.385	-4.033	5.194	-0.9960
	-1.385	-4.033	-5.194	0.1262E-06
0.8600	-1.364	-3.824	6.067	-0.9961
	-1.364	-3.824	-6.067	0.1316E-06
0.8200	-1.344	-3.620	6.789	-0.9962
	-1.344	-3.620	-6.789	0.6273E-06
0.7800	-1.325	-3.419	7.405	-0.9963
	-1.325	-3.419	-7.405	0.1573E-06
0.7200	-1.298	-3.126	8.182	-0.9965
	-1.298	-3.126	-8.182	-0.2419E-06
0.6400	-1.264	-2.747	9.016	-0.9967
	-1.264	-2.747	-9.016	-0.6486E-07
0.5600	-1.233	-2.380	9.675	-0.9969
	-1.233	-2.380	-9.675	0.6588E-07
0.4800	-1.205	-2.026	10.20	-0.9971
	-1.205	-2.026	-10.20	-0.6353E-07
0.4000	-1.180	-1.681	10.60	-0.9973
	-1.180	-1.681	-10.60	-0.2348E-07
0.3200	-1.158	-1.347	10.91	-0.9975
	-1.158	-1.347	-10.91	-0.1736E-07
0.2200	-1.134	-0.9407	11.18	-0.9977
	-1.134	-0.9406	-11.18	-0.9304E-08
0.1000E 00	-1.108	-0.4678	11.33	-0.9979
	-1.108	-0.4678	-11.33	-0.2254E-07
-0.2000E-01	-1.087	-0.9100E-02	11.32	-0.9982
	-1.087	-0.9078E-02	-11.32	-0.3505E-07
-0.1600	-1.066	0.5107	11.11	-0.9984
	-1.066	0.5107	-11.11	-0.3376E-07
-0.3200	-1.046	1.089	10.59	-0.9986
	-1.046	1.089	-10.59	-0.3322E-07
-0.5000	-1.028	1.722	9.618	-0.9988
	-1.028	1.722	-9.618	-0.2301E-07
-0.7200	-1.012	2.477	7.651	-0.9990
	-1.012	2.477	-7.651	-0.4846E-07
-0.9200	-1.001	3.151	4.296	-0.9991
	-1.001	3.151	-4.296	-0.3500E-07

U	UU	DU	DV
11.53	-0.2813E-01	1.848	-2.742
7.034	-0.6602E-01	2.875	-2.401
5.471	-0.9076E-01	3.515	-2.082
4.568	-0.1060	3.982	-1.772
3.950	-0.1122	4.338	-1.472
3.488	-0.1098	4.610	-1.184
3.122	-0.9855E-01	4.813	-0.9088
2.819	-0.7888E-01	4.958	-0.6465
2.561	-0.5043E-01	5.050	-0.3989
2.336	-0.1351E-01	5.094	-0.1675
2.138	0.3156E-01	5.091	0.4345E-01
1.961	0.8462E-01	5.045	0.2323
1.799	0.1461	4.956	0.3986
1.649	0.2159	4.823	0.5390

1.508	0.2943	4.646	0.6501
1.373	0.3817	4.419	0.7268
1.247	0.4746	4.149	0.7609
1.139	0.5629	3.868	0.7509
1.052	0.6389	3.606	0.7091
0.9798	0.7037	3.363	0.6467
0.9168	0.7599	3.131	0.5680
0.8598	0.8098	2.906	0.4747
0.8088	0.8523	2.692	0.3717
0.7638	0.8873	2.495	0.2639
0.7218	0.9148	2.305	0.1480
0.6828	0.9382	2.123	0.2649E-01
0.6468	0.9532	1.951	-0.9815E-01
0.6138	0.9646	1.792	-0.2232
0.5838	0.9682	1.646	-0.3460
0.5538	0.9669	1.500	-0.4776
0.5238	0.9595	1.354	-0.6177
0.4938	0.9453	1.211	-0.7663
0.4668	0.9288	1.083	-0.9068
0.4428	0.9062	0.9725	-1.037
0.4188	0.8784	0.8648	-1.171
0.3948	0.8449	0.7608	-1.309
0.3708	0.8055	0.6611	-1.451
0.3468	0.7603	0.5666	-1.594
0.3228	0.7093	0.4780	-1.739
0.2988	0.6529	0.3958	-1.884
0.2748	0.5917	0.3208	-2.027
0.2508	0.5265	0.2535	-2.168
0.2268	0.4584	0.1944	-2.305
0.2028	0.3890	0.1436	-2.435
0.1788	0.3200	0.1013	-2.558
0.1578	0.2595	0.7108E-01	-2.657
0.1398	0.2113	0.5018E-01	-2.735
0.1218	0.1661	0.3356E-01	-2.806
0.1038	0.1250	0.2088E-01	-2.868
0.8578E-01	0.8872E-01	0.1175E-01	-2.922
0.6778E-01	0.5820E-01	0.5675E-02	-2.967
0.4978E-01	0.3411E-01	0.2098E-02	-3.001
0.3178E-01	0.1705E-01	0.4079E-03	-3.025
0.1378E-01	0.7755E-02	-0.7381E-04	-3.039
0.0000	-0.1527E-02	0.8934E-05	-3.042
-0.1378E-01	0.7755E-02	0.9418E-04	-3.039
-0.3178E-01	0.1706E-01	-0.3873E-03	-3.025
-0.4978E-01	0.3411E-01	-0.2077E-02	-3.001
-0.6778E-01	0.5820E-01	-0.5655E-02	-2.966
-0.8578E-01	0.8872E-01	-0.1173E-01	-2.922
-0.1038	0.1250	-0.2086E-01	-2.868
-0.1218	0.1661	-0.3354E-01	-2.806
-0.1398	0.2113	-0.5016E-01	-2.735
-0.1578	0.2595	-0.7106E-01	-2.657
-0.1788	0.3200	-0.1013	-2.558
-0.2028	0.3890	-0.1436	-2.435
-0.2268	0.4584	-0.1944	-2.305
-0.2508	0.5265	-0.2535	-2.168
-0.2748	0.5917	-0.3208	-2.027
-0.2988	0.6529	-0.3958	-1.884
-0.3228	0.7093	-0.4779	-1.739
-0.3468	0.7603	-0.5666	-1.594
-0.3708	0.8055	-0.6611	-1.451
-0.3948	0.8449	-0.7607	-1.309
-0.4188	0.8784	-0.8648	-1.171
-0.4428	0.9062	-0.9725	-1.037
-0.4668	0.9288	-1.083	-0.9068

-0.4938	0.9453	-1.211	-0.7663
-0.5238	0.9595	-1.355	-0.6177
-0.5538	0.9669	-1.500	-0.4776
-0.5838	0.9682	-1.646	-0.3460
-0.6138	0.9646	-1.792	-0.2232
-0.6468	0.9532	-1.951	-0.9815E-01
-0.6828	0.9382	-2.123	0.2649E-01
-0.7218	0.9148	-2.305	0.1480
-0.7638	0.8873	-2.495	0.2639
-0.8088	0.8523	-2.692	0.3717
-0.8598	0.8098	-2.905	0.4747
-0.9168	0.7599	-3.131	0.5680
-0.9798	0.7037	-3.363	0.6467
-1.052	0.6389	-3.606	0.7091
-1.139	0.5629	-3.868	0.7509
-1.247	0.4746	-4.149	0.7609
-1.373	0.3817	-4.419	0.7268
-1.508	0.2943	-4.646	0.6501
-1.649	0.2159	-4.823	0.5390
-1.799	0.1462	-4.956	0.3986
-1.961	0.8462E-01	-5.045	0.2323
-2.138	0.3156E-01	-5.091	0.4345E-01
-2.336	-0.1351E-01	-5.094	-0.1675
-2.561	-0.5043E-01	-5.050	-0.3989
-2.819	-0.7888E-01	-4.958	-0.6465
-3.122	-0.9856E-01	-4.813	-0.9088
-3.488	-0.1098	-4.610	-1.184
-3.950	-0.1122	-4.338	-1.472
-4.568	-0.1060	-3.982	-1.772
-5.471	-0.9076E-01	-3.515	-2.082
-7.034	-0.6602E-01	-2.875	-2.401
-11.53	-0.2813E-01	-1.848	-2.742

```
R =     0.5000      DC =     1.500      DIP =    0.0000
SM =    0.1000E 06  PR =    0.2500      DPC =    -1.000      DPG =    0.0000
TDS =    30.00      ERSU =   0.1000E-01 ERSL =    0.1000E-02 ERDP =    0.1000E-0
KNORM =   1.000     SNORM =   1.000     DNORM =    0.7500E-06
NSURF =   116
NSLIT =    36
NIT =      4 NSUPER =       94900
KP =    1.255     -0.7384E-01  KN =     1.265      0.7382E-01
```

X	XX/XXN	DX/DXN	DY/DYN	YY/XY
0.9800	-0.4233	-4.684	0.9576	-0.9979
	-1.824	-4.393	-4.084	-0.1571E-02
0.9400	-0.7790	-4.621	2.874	-0.9978
	-1.494	-4.137	-5.799	-0.1556E-02
0.9000	-0.9106	-4.517	4.194	-0.9976
	-1.390	-3.915	-6.922	-0.1536E-02
0.8600	-0.9939	-4.393	5.260	-0.9975
	-1.333	-3.711	-7.794	-0.1511E-02
0.8200	-1.057	-4.254	6.171	-0.9974
	-1.296	-3.515	-8.516	-0.1482E-02
0.7800	-1.109	-4.103	6.972	-0.9972
	-1.269	-3.325	-9.133	-0.1447E-02
0.7400	-1.155	-3.943	7.689	-0.9971
	-1.249	-3.141	-9.671	-0.1408E-02
0.7000	-1.195	-3.774	8.336	-0.9970
	-1.233	-2.959	-10.15	-0.1364E-02
0.6600	-1.232	-3.597	8.924	-0.9969
	-1.221	-2.781	-10.57	-0.1315E-02
0.6200	-1.266	-3.413	9.461	-0.9968
	-1.210	-2.604	-10.95	-0.1262E-02
0.5800	-1.297	-3.223	9.952	-0.9966
	-1.201	-2.430	-11.29	-0.1204E-02
0.5400	-1.326	-3.026	10.40	-0.9965
	-1.194	-2.257	-11.59	-0.1142E-02
0.5000	-1.354	-2.824	10.81	-0.9964
	-1.186	-2.086	-11.87	-0.1076E-02
0.4400	-1.385	-2.511	11.36	-0.9963
	-1.184	-1.831	-12.23	-0.9686E-03
0.3600	-1.427	-2.079	11.96	-0.9961
	-1.175	-1.495	-12.62	-0.8137E-03
0.2800	-1.462	-1.631	12.44	-0.9960
	-1.168	-1.161	-12.92	-0.6463E-03
0.1800	-1.488	-1.056	12.86	-0.9959
	-1.167	-0.7472	-13.18	-0.4234E-03
0.0000	-1.488	-0.6459E-04	13.16	-0.9958
	-1.185	-0.3553E-04	-13.37	0.6330E-07
-0.1800	-1.488	1.055	12.86	-0.9959
	-1.167	0.7472	-13.18	0.4235E-03
-0.2800	-1.462	1.631	12.44	-0.9960
	-1.168	1.161	-12.92	0.6464E-03
-0.3600	-1.427	2.079	11.96	-0.9961
	-1.175	1.494	-12.62	0.8137E-03
-0.4400	-1.385	2.511	11.36	-0.9963
	-1.184	1.831	-12.23	0.9687E-03
-0.5000	-1.354	2.824	10.81	-0.9964
	-1.186	2.086	-11.87	0.1076E-02
-0.5400	-1.326	3.025	10.40	-0.9965
	-1.194	2.257	-11.59	0.1142E-02
-0.5800	-1.297	3.222	9.952	-0.9966
	-1.201	2.430	-11.29	0.1204E-02

-0.6200	-1.266	3.413	9.451	-0.9968
	-1.210	2.604	-10.95	0.1262E-02
-0.6600	-1.232	3.597	8.924	-0.9969
	-1.221	2.781	-10.57	0.1315E-02
-0.7000	-1.195	3.773	9.336	-0.9970
	-1.233	2.959	-10.15	0.1364E-02
-0.7400	-1.155	3.942	7.689	-0.9971
	-1.249	3.141	-9.671	0.1408E-02
-0.7800	-1.109	4.103	6.972	-0.9972
	-1.269	3.325	-9.133	0.1447E-02
-0.8200	-1.057	4.253	6.171	-0.9974
	-1.296	3.515	-8.516	0.1481E-02
-0.8600	-0.9940	4.392	5.260	-0.9975
	-1.333	3.711	-7.794	0.1511E-02
-0.9000	-0.9106	4.517	4.194	-0.9976
	-1.390	3.916	-6.922	0.1535E-02
-0.9400	-0.7790	4.621	2.874	-0.9978
	-1.494	4.137	-5.799	0.1556E-02
-0.9800	-0.4232	4.683	0.9576	-0.9979
	-1.823	4.393	-4.034	0.1571E-02

U	UU	DU	DV
11.48	-0.2859E-02	0.1003	-5.667
5.902	-0.1455E-01	0.2052	-5.612
5.345	-0.3561E-01	0.3587	-5.552
4.555	-0.6169E-01	0.5484	-5.473
4.075	-0.8942E-01	0.7251	-5.333
3.737	-0.1180	0.8989	-5.232
3.479	-0.1465	1.069	-5.174
3.269	-0.1754	1.239	-5.056
3.092	-0.2042	1.407	-4.929
2.942	-0.2321	1.571	-4.794
2.813	-0.2586	1.730	-4.655
2.699	-0.2838	1.886	-4.510
2.594	-0.3082	2.042	-4.355
2.493	-0.3315	2.197	-4.192
2.411	-0.3526	2.347	-4.024
2.330	-0.3722	2.495	-3.849
2.255	-0.3897	2.639	-3.667
2.183	-0.4050	2.784	-3.475
2.114	-0.4188	2.927	-3.272
2.051	-0.4299	3.062	-3.069
1.991	-0.4376	3.194	-2.860
1.931	-0.4429	3.327	-2.633
1.874	-0.4462	3.455	-2.401
1.820	-0.4454	3.577	-2.165
1.766	-0.4414	3.699	-1.913
1.712	-0.4336	3.818	-1.644
1.661	-0.4235	3.923	-1.375
1.613	-0.4090	4.029	-1.107
1.565	-0.3906	4.127	-0.8250
1.517	-0.3679	4.219	-0.5239
1.469	-0.3407	4.305	-0.2188
1.421	-0.3086	4.384	0.1050
1.373	-0.2716	4.455	0.4422
1.325	-0.2294	4.517	0.7921
1.277	-0.1820	4.567	1.154
1.232	-0.1341	4.604	1.503
1.190	-0.8435E-01	4.629	1.839
1.148	-0.3106E-01	4.641	2.179
1.105	0.2569E-01	4.642	2.525

1.064	0.8561E-01	4.631	2.877
1.022	0.1485	4.606	3.232
0.9799	0.2138	4.569	3.590
0.9378	0.2814	4.517	3.948
0.8958	0.3506	4.451	4.306
0.8538	0.4211	4.369	4.661
0.8118	0.4923	4.273	5.013
0.7698	0.5638	4.162	5.360
0.7278	0.6351	4.035	5.699
0.6858	0.7056	3.894	6.031
0.6438	0.7750	3.738	6.352
0.5988	0.8462	3.555	6.683
0.5509	0.9206	3.341	7.019
0.5029	0.9917	3.110	7.336
0.4518	1.061	2.847	7.649
0.3948	1.134	2.531	7.966
0.3298	1.210	2.140	8.239
0.2478	1.303	1.626	8.612
0.0000	1.251	-0.7575E-04	9.015
-0.2478	1.303	-1.626	8.612
-0.3299	1.210	-2.140	8.289
-0.3949	1.134	-2.531	7.966
-0.4518	1.061	-2.847	7.649
-0.5028	0.9917	-3.111	7.336
-0.5508	0.9206	-3.342	7.019
-0.5988	0.8462	-3.555	6.683
-0.6438	0.7750	-3.738	6.352
-0.6858	0.7056	-3.895	6.031
-0.7278	0.6351	-4.036	5.699
-0.7698	0.5638	-4.162	5.360
-0.8118	0.4923	-4.274	5.013
-0.8538	0.4211	-4.370	4.661
-0.8958	0.3506	-4.451	4.306
-0.9378	0.2814	-4.517	3.948
-0.9793	0.2139	-4.569	3.590
-1.022	0.1485	-4.607	3.232
-1.064	0.8561E-01	-4.631	2.877
-1.106	0.2569E-01	-4.642	2.525
-1.148	-0.3105E-01	-4.641	2.179
-1.190	-0.8435E-01	-4.628	1.838
-1.232	-0.1341	-4.604	1.503
-1.277	-0.1820	-4.567	1.154
-1.325	-0.2294	-4.517	0.7921
-1.373	-0.2716	-4.456	0.4421
-1.421	-0.3086	-4.385	0.1050
-1.469	-0.3407	-4.303	-0.2188
-1.517	-0.3679	-4.219	-0.5289
-1.565	-0.3906	-4.127	-0.8250
-1.613	-0.4090	-4.030	-1.107
-1.661	-0.4235	-3.928	-1.375
-1.712	-0.4335	-3.818	-1.644
-1.766	-0.4414	-3.698	-1.913
-1.820	-0.4454	-3.577	-2.165
-1.874	-0.4462	-3.453	-2.401
-1.931	-0.4429	-3.327	-2.633
-1.991	-0.4375	-3.194	-2.850
-2.051	-0.4298	-3.062	-3.059
-2.114	-0.4188	-2.927	-3.272
-2.183	-0.4050	-2.784	-3.475
-2.255	-0.3897	-2.633	-3.667

-2.330	-0.3722	-2.495	-3.849
-2.411	-0.3526	-2.347	-4.024
-2.498	-0.3315	-2.197	-4.192
-2.594	-0.3082	-2.043	-4.355
-2.699	-0.2839	-1.886	-4.510
-2.813	-0.2586	-1.730	-4.655
-2.942	-0.2321	-1.571	-4.794
-3.092	-0.2042	-1.407	-4.923
-3.263	-0.1754	-1.238	-5.056
-3.479	-0.1466	-1.069	-5.174
-3.737	-0.1180	-0.8991	-5.282
-4.076	-0.8942E-01	-0.7252	-5.383
-4.556	-0.6169E-01	-0.5485	-5.473
-5.345	-0.3561E-01	-0.3688	-5.552
-6.902	-0.1455E-01	-0.2053	-5.612
-11.49	-0.2960E-02	-0.1004	-5.657